

# Excel Funktionen

Kategorie Text



# Verkettungsoperator: &

- ▶ Verkettung (Konkatenieren) von Werten
  - Zellbezüge
  - Literale Konstanten

	A	B	C
1	Puter	Com	5%

Anweisung	Ergebnis
=B1 & A1	ComPuter
= "Ich heiße " & A1 & ", " & B1 & " " & A1 & "!"	Ich heiße Puter, Com Puter!
= "Ich erhalte " & C1 & " Rabatt."	Ich erhalte 0,05 Rabatt. *)

\*) Verwenden Sie die TEXT-Funktion, um den Inhalt von C1 als Prozentzeichenkette formatiert zu erhalten.

# TEXT-Funktion

- ▶ Liefert Werte als Zeichenkette in einem bestimmten Format \*)

	A	B	C
1	09.10.2015 16:25:30	0,048	2500

Anweisung	Ergebnis
=Text( A1; "tttt" )	Freitag
=Text( A1; "MMM" )	Okt
=Text( A1; "mm" )	25
=Text( B1; "0,00%" )	4,8%
=Text( B1; "0%" )	5%
=Text( C1; "#.##0,00" )	2.500,00
=Text( C1; "€ #.##0" )	€ 2.500

\*) Recherchieren Sie die Möglichkeiten zur Definition einer Formatmaske in der Excel-Hilfe.

# GROSS- bzw. KLEIN-Funktion

- ▶ Liefert eine Zeichenkette in Groß- bzw. Kleinbuchstaben

Anweisung	Ergebnis
=Gross( "Mustermann" )	MUSTERMANN
=Klein( "Mustermann" )	mustermann
=Gross( Text( Datwert( "05.12.2015"); "tttt" ) )	SAMSTAG *)

\*) Die DATWERT-Funktion ist in den Unterlagen zu den Datumsfunktionen näher erläutert.

# LÄNGE-Funktion

- ▶ Liefert die Anzahl der Zeichen einer Zeichenkette
  - Berücksichtigt führende und nachgestellte Leerzeichen

Anweisung	Ergebnis
=Länge( "Max" )	3
=Länge( " Max" )	4
=Länge( " Max " )	5

# GLÄTTEN-Funktion

- ▶ Liefert die Zeichenkette um Leerzeichen bereinigt
  - Alle führenden Leerzeichen werden entfernt
  - Alle nachgestellten Leerzeichen werden entfernt
  - Einzelne Leerzeichen innerhalb der Zeichenkette bleiben erhalten
  - Aufeinanderfolgende Leerzeichen innerhalb der Zeichenkette werden auf **ein** Leerzeichen reduziert.

Anweisung	Ergebnis
=Glätten ( " Max Mustermann " )	Max Mustermann
=Glätten( " M a x " )	M a x
=Länge( Glätten( " M a x " ) )	5

# LINKS- bzw. RECHTS-Funktion

- ▶ Liefert eine bestimmte Anzahl von Zeichen am Beginn oder am Ende einer Zeichenkette
- ▶ Zwei Argumente
  - Zeichenkette
  - Anzahl Zeichen
    - Anzahl der Zeichen, die vom Beginn (Links) bzw. vom Ende (Rechts) der Zeichenkette zurückgegeben werden

# LINKS- bzw. RECHTS-Funktion

Anweisung	Ergebnis
=Links( "Mustermann"; 4 )	Must
=Links( " Mustermann"; 4)	_Mus (Leerzeichen am Beginn)
=Links( Glätten( " Mustermann" ); 4 )	Must
=Links( "Mustermann"; 100 )	Mustermann (bis zum Ende der Zeichenkette)
=Rechts( "Mustermann"; 4 )	mann
=Rechts( "Mustermann "; 4)	ann_ (Leerzeichen am Ende)
=Rechts( Glätten( "Mustermann " ); 4 )	mann
=Rechts( "Mustermann"; 100 )	Mustermann (bis zum Anfang der Zeichenkette)

# TEIL-Funktion

- ▶ Liefert den Teil einer Zeichenkette, der an einer bestimmten Position beginnt und eine bestimmte Anzahl von Zeichen lang ist
- ▶ Drei Argumente
  - Zeichenkette
  - Erstes Zeichen
    - Position des Zeichens, bei dem mit der Extrahierung begonnen wird
    - Index ist 1-basiert
  - Anzahl Zeichen
    - Anzahl der zu extrahierenden Zeichen
    - Ist die Anzahl grösser als der verbleibende Text, so wird bis zum Ende der Zeichenkette extrahiert

# TEIL-Funktion

Anweisung	Ergebnis
=Teil( "Mustermann"; 3; 4 )	ster
=Teil( " Mustermann"; 3; 4)	uste
=Teil ( Glätten( " Mustermann" ); 3; 4 )	ster
=Teil( "Mustermann"; 3; 100)	stermann (bis zum Ende der Zeichenkette)
=Teil( "Mustermann"; 3; Länge( "Mustermann" ) )	stermann (Gesamtlänge ist lang genug, um bis zum Ende der Zeichenkette zu extrahieren)

# SUCHEN- bzw. FINDEN-Funktion

- ▶ Liefert die Position (1-basiert) eines Zeichens (bzw. einer Zeichenkette) innerhalb einer Zeichenkette
- ▶ Berücksichtigung der Groß- / Kleinschreibung
  - SUCHEN: Wird nicht berücksichtigt
  - FINDEN: Wird berücksichtigt
- ▶ Drei Argumente
  - Suchbegriff
  - Zeichenkette
  - Erstes Zeichen (optional)
    - Position an der mit der Suche begonnen wird (1-basiert)
- ▶ Keine Übereinstimmung
  - Ist der Suchbegriff in der Zeichenkette nicht enthalten, so wird der Fehler #WERT! zurückgeliefert

# SUCHEN- bzw. FINDEN-Funktion

Anweisung	Ergebnis
=Suchen( "m"; "Max" )	1
=Finden( "m"; "Max" )	#WERT! (Groß- / Kleinschreibung)
=Finden( "M"; "Max" )	1
=Suchen( "M"; "Mustermann"; 2)	7
=Finden( "M"; "Mustermann"; 2)	#WERT! (Groß- / Kleinschreibung)
=Suchen( " "; "Max Mustermann" )	4 (Position des Leerzeichens)
=Suchen( " "; "Max Mustermann") + 1	5 (Position des Zeichens nach dem Leerzeichen)
=Suchen( " "; "Max Muster Mann"; Suchen(" "; "Max Muster Mann") + 1 )	11 (Position des zweiten Leerzeichens, da die Suche eine Position nach dem ersten Leerzeichen beginnt)
=WennFehler( Suchen( "y"; "Max" ); -1)	-1 *)

\*) Die WENNFehler-Funktion ist in den Unterlagen zur Kategorie Logik erläutert.

# Kombinationen

- ▶ Auslesen bis zum ersten Leerzeichen

	A
1	Max Mustermann

- Ermitteln der Position des ersten Leerzeichens  
=Suchen( " "; A1) → 4
- Variante mit LINKS-Funktion  
=Links( A1; Suchen( " "; A1) - 1 )
- Variante mit TEIL-Funktion  
=Teil( A1; 1; Suchen( " "; A1 ) - 1 )

# Kombinationen

- ▶ Auslesen nach dem ersten Leerzeichen bis zum Ende der Zeichenkette

	A
1	Max Mustermann

- Ermitteln der Position des ersten Leerzeichens  
=Suchen( " "; A1) → 4
- Variante mit RECHTS-Funktion \*)  
=Rechts( A1; Länge( A1 ) - Suchen( " "; A1 ) )
- Variante mit TEIL-Funktion  
=Teil( A1; Suchen( " "; A1 ) +1; Länge( A1 ) )

\*) Beachten Sie, dass die RECHTS-Funktion die Anzahl der Zeichen vom Ende aus zählt.

# Kombinationen

- ▶ Auslesen zwischen erstem Leerzeichen und Strichpunkt

	A
1	Max Mustermann; Innsbruck

- Ermitteln der Position des ersten Leerzeichens  
=Suchen( " "; A1) → 4
- Ermitteln der Position des ersten Strichpunkts  
=Suchen( ";"; A1) → 15
- Ergebnis mit TEIL-Funktion \*)  
=Teil( A1; Suchen( " "; A1) + 1;  
Suchen( ";"; A1) - Suchen( " "; A1 ) - 1 )

\*) Beim Ermitteln der Anzahl der Zeichen rechnen Sie: 15 - 4 - 1 → Länge von Mustermann: 10

# Kombinationen

- ▶ Auslesen zwischen erstem und zweitem Leerzeichen

	A
1	Max Mustermann Innsbruck

- Ermitteln der Position des ersten Leerzeichens  
=Suchen( " "; A1) → 4
- Ermitteln der Position des zweiten Leerzeichens \*)  
=Suchen( " "; A1; Suchen( " "; A1 ) + 1 ) → 15
- Ergebnis mit TEIL-Funktion  
=Teil( A1; Suchen( " "; A1) + 1;  
Suchen( " "; A1; Suchen( " "; A1 ) + 1 ) - Suchen( " "; A1 ) - 1 )

\*) Suche nach dem zweiten Leerzeichen beginnt an der Position nach dem ersten Leerzeichen